In-class Exercises: More Advanced XPath and XQuery

Our ultimate goal is to find the sid of every student who gave an answer to every True-False question on the quiz, and to produce output in this format:

```
<COMPLETE>
  <SID>12345</SID>
  <SID>23456</SID>
  <SID>34567</SID>
</COMPLETE>
```

We'll develop the answer in several steps.

1. First, find the sid of every student. Any output format is fine.

   Solution:

   ```
   fn:doc("class.xml")/ClassResponses/Student/@sid
   ```

2. Now do the same thing, but make the output be a sequence of SID elements, each with a student number inside. Example: `<SID>998801234</SID>`

   Solution:

   ```
   for $student in fn:doc("class.xml")/ClassResponses/Student
   return
     <SID>
       { data($student/@sid) }
     </SID>
   ```

3. Write a query to find the QID of every quiz question. Any output format is fine. (We'll integrate this with the information about students later.)

   Solution:

   ```
   let $qdoc := fn:doc("quiz.xml")
   for $question in $qdoc/Quiz/Question/@QID
   return $question
   ```
4. Modify your last query to find QIDs of only the True-False questions.

Solution:

```xml
let $qdoc := fn:doc("quiz.xml")
let $bdoc := fn:doc("bank.xml")
let $banktfquestions := $bdoc//TFQuestion/@QID
for $question in $qdoc/Quiz/Question/@QID
    where $question = $banktfquestions
    return $question
```

5. Go back to your query that prints out all the SIDs. Extend it by adding a where clause that will later let us pick only
   the right students to include. For now, make that where clause always succeed. Add in code that stores the QIDs of
   the True-False questions in a variable. (We'll use that when we write the proper where clause.)

Solution:

```xml
let $qdoc := fn:doc("quiz.xml")
let $bdoc := fn:doc("bank.xml")
let $cdoc := fn:doc("class.xml")
let $banktfquestions := $bdoc//TFQuestion/@QID
let $quiztfquestions :=
    for $question in $qdoc/Quiz/Question/@QID
    where $question = $banktfquestions
    return $question
for $student in $cdoc/ClassResponses/Student
    where true()
    return
    <SID>
    { data($student/@sid) }
</SID>
```
6. Modify the where clause so that only students who’ve answered every True-False question are included in the output.

Solution:

```xml
let $qdoc := fn:doc("quiz.xml")
let $bdoc := fn:doc("bank.xml")
let $cdoc := fn:doc("class.xml")
let $banktfquestions := $bdoc//TFQuestion/@QID
let $quiztfquestions :=
  for $question in $qdoc/Quiz/Question/@QID
  where $question = $banktfquestions
  return $question
for $student in $cdoc/ClassResponses/Student
let $studentanswered := $student/QuestionResponse/@QID
where every $q in $quiztfquestions satisfies $q = $studentanswered
return
  <SID>
  { data($student/@sid) }
  </SID>
```

7. Finally, modify your code so that all the SID elements are wrapped up in a COMPLETE element.

Solution:

```xml
<COMPLETE>

  { 
    let $qdoc := fn:doc("quiz.xml")
    let $bdoc := fn:doc("bank.xml")
    let $cdoc := fn:doc("class.xml")
    let $banktfquestions := $bdoc//TFQuestion/@QID
    let $quiztfquestions :=
      for $question in $qdoc/Quiz/Question/@QID
      where $question = $banktfquestions
      return $question
    for $student in $cdoc/ClassResponses/Student
    let $studentanswered := $student/QuestionResponse/@QID
    where every $q in $quiztfquestions satisfies $q = $studentanswered
    return
      <SID>
      { data($student/@sid) }
      </SID>
  }
</COMPLETE>
```